# $Id: PKGBUILD 83677 2013-02-04 11:05:12Z spupykin $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: kevin <kevin@archlinux.org>
# Contributor: iztok pizorn <pizorn@gmail.com>

pkgname=fldiff
pkgver=1.1
pkgrel=7
pkgdesc="graphical diff program"
arch=(i686 x86_64)
url="http://www.easysw.com/~mike/fldiff/"
license=('GPL')
install=fldiff.install
depends=('diffutils' 'fltk' 'xdg-utils')
source=(http://www.easysw.com/~mike/fldiff/$pkgname-$pkgver-source.tar.gz
	build-fix.patch)
md5sums=('12897c92106fb5d665210a2e82f5cf25'
         'ea7258e07544b81561c6d76a41c184e7')

build() {
  cd $srcdir/$pkgname-$pkgver
  patch -p1 <$srcdir/build-fix.patch
  ./configure --prefix=/usr
  make
}

package() {
  cd $srcdir/$pkgname-$pkgver
  make BUILDROOT=$pkgdir install install-desktop
  mv `find $pkgdir/ -name Development -type d` $pkgdir/usr/share/applications
}
